


テーマ	MPC-1000 パルス発生例
使用機器	MPC-1000 他

■機器構成



■サンプルプログラム 1

- 原点復帰から始めます。
- 原点復帰開始時に原点LSがオンしていればCW方向へ一定量退避移動します。
- 原点検出は、パルス出力→LSオン待ち→パルス停止→現在位置クリア
- 原点復帰後 CW方向へ相対移動を繰り返し→絶対座標移動で0に戻ります。
- “A”コマンドは加減速テーブルを算出してCPU内臓フラッシュromに保存します。加減速テーブルの作成には多少時間がかかります。また、フラッシュromの書き込み回数には限界があります(10万回未満などとされています)。ですからプログラム中で“A”の変更は避けてください。“A”は実行時にフラッシュROMの既存データを参照して、同じなら再演算・再保存は行いません。
- “C”や“V”も実行に時間がかかります。プログラム中での頻繁な使用には注意してください。

```

PGA "D" 0          /* "D" 0=CW/CCW(default), 1=PULSE/DIR
PGA "A" 5000       /* "A" acceleration/deceleration 500~12000 pps
WAIT SW(204)==1   /* wait for ready

GOSUB *PGA_HOME

DO
  FOR I=1 TO 10
    PGA "F" I       /* "F" speed 10(max)~0(min)
    PGA "R" 1000    /* "R" move to relative coordinate
    WAIT SW(204)==1
    PGA "C"         /* "C" get current position
    PRINT V_PGA     /* display position. V_PGA is a reserved variable
  NEXT I

  PGA "F" 10       /* change speed (10=max speed)
  PGA "M" 0        /* "M" move to absolute coordinate
  WAIT SW(204)==1 /* wait for ready
  PGA "C"

```

```

PRINT V_PGA
LOOP
*PGA_HOME                               /* PGA home position detection
IF SW(198)==1 THEN                       /* if origin limit switch is on
  PGA "F" 5                               /* speed
  PGA "R" 1000                             /* PGA moves in the counter direction with the ORG
  WAIT SW(204)==1                         /* wait for ready
END_IF
PGA "G" -4000                             /* "G" constant speed drive. CCW 4Kpps
WAIT SW(198)==1                           /* wait for the ORG switch until on
OFF PGA                                    /* PGA pulse stop immediately
TIME 50                                    /* need
PGA "H" 0                                  /* "H" set of the current position
PGA "V"                                    /* "V" get PGA firmware version
PRINT "PGA version=" V_PGA               /* display version
PGA "C"
PRINT V_PGA
RETURN

```

実行結果 (FTMW の表示)

```

PGA version= 100215      ←PGAのファームウェアのバージョンです
0
1000
2000
3000
4000
5000
6000
7000
8000
9000
10000
0
1000

```

■ サンプルプログラム 2

- 一定パルスレート、定量パルス発生です。
- 加減速はありません。
- "C"による現在位置の取得はできません。

```

PGA "S" 5000                               /* set the pulse rate of "P", "W" command
DO
  PGA "P" 5000                             /* "P" pulse out. CW 5000 without acc/dec.
  WAIT SW(204)==1                         /* wait for ready
  PGA "C"                                   /* "C" command is invalid to "P"
  PRINT V_PGA                              /* display
  PGA "P" -5000                             /* pulse out. CCW 5000 without acc/dec.
  WAIT SW(204)==1                         /* wait for ready
  PGA "C"
  PRINT V_PGA
LOOP

```

実行結果 (FTMW の表示)

```

0
0
0

```

■ サンプルプログラム 3

- パルス発生中、押しボタン SW(192) でパルスレートを変更します。ボタンを押さなければ変化せずに規定量を移動します。

```

PGA "S" 1000                               /* set the pulse rate of "P", "W" command
PGA "P" 5000                               /* "P" pulse out. CW 5000 without acc/dec.
DO
  IF SW(192)==1 THEN                       /* push switch
    PGA "S" 2000
  END_IF
  IF SW(204)==1 THEN                       /* PGA ready
    BREAK                                  /* exit from DO~LOOP
  END_IF
LOOP
PRINT "FINISHED"

```

■コマンドの補足

PGA "G" pps : PPS 指定パルス発生 (20~9000pps)
PGA "S" pps : パルスレート設定 (20~9000pps)
PGA "W" duty : PWM (40~970/1000)
PGA "P" pls : パルス数指定パルス発生 (+/-8000000)
PGA "A" pps : 加減速テーブル生成 (500~12000pps)
PGA "F" f : 速度選択 (10~0)
PGA "R" pls : 加減速パルス発生 相対 (+/-8000000)
PGA "M" pos : 加減速パルス発生 座標 (+/-8000000)
PGA "H" pos : 現在位置設定 (+/-8000000)
PGA "D" n : パルス方式 (0: デフォルト 2PLS 1: 方向指示)
PGA "C" : 現在位置取得
PGA "V" : バージョン取得

"G"はパルスレートを指定してパルス発生をします。無限パルス発生です。

"P"はパルス数を指定してパルス発生をします。

"P" "G"に加減速はありません。


"S"は "P", "W" のパルスレートを設定します。

"A"は "R", "M" の加減速・最高速度を作成します。台形のみみたいなイメージです。加速距離は最高速の 1/10。

"F"は "R", "M" で動かすとき、"A" で作成した山の何合目まで登るかを決めます。

"A"の実行(再演算)は時間がかかります。なので、プログラムの冒頭で "A" をして、"R", "M" の速度を変更したいときは "F" します。

"F"は "P", "W", "G" に影響しません。



"D"はパルス出力方式の選択で、モータドライバの入力仕様に合わせます。

ステップモータ、サーボモータのドライバのパルス入力には CW/CCW 方式と方向指示方式があります。

(メーカーによって呼び方が違い CW/CCW を 2 パルス方式などということもあります)

ドライバのパルス入力は 2 つありますが、それを正転信号と逆転信号とするか、片側をパルス信号、他方を方向切替信号とするかです。

上記のサンプルでもやっていますが、"A", "R", "M", "P" の次の行で WAIT SW (204) == 1 をしています。

これはコマンドの実行完了を待ちます。これが無いとパルスを出し終わらないのにプログラムが先に進んでしまうからです(そういう使い方もあります)。

サンプルは PGA なので SW (204) が READY です。PGB なら SW (206) が READY です。

PGA, PGB は簡易的なパルス発生です。それなりに誤差もあります。

最高速実測値

PGB "A" 11000 のとき 11.2kpps

PGB "A" 12000 のとき 12.3kpps